

# SEMI-SIMPLICIAL TYPES

Astra Kolomatskaia (*she/her*)

20 November 2022

## AN INTRODUCTION TO THE PROBLEM

Let  $\Delta^n$  denote the standard simplex in  $\mathbb{R}^n$  – this is defined to be the convex hull of standard basis elements  $e_1, \dots, e_n$  together with the point  $e_0 = \mathbf{0}$ .  $\Delta^0$  is a single point,  $\Delta^1$  is a line,  $\Delta^2$  is a triangle,  $\Delta^3$  is a tetrahedron, etc.

Given a topological space  $X$ , we can define  $S[X]_n$  to be the collection of maps  $\Delta^n \rightarrow X$ . Thus  $S[X]_0$  represents the points in  $X$ ,  $S[X]_1$  represents the lines in  $X$ , and  $S[X]_2$  represents the triangles in  $X$ , etc.

The collections  $S[X]_n$  are related by **face maps** that encode information about the boundary of a simplex. We define affine maps  $\delta_k : \Delta^{n-1} \rightarrow \Delta^n$ , for  $k \in \{0, \dots, n\}$ , by mapping the points  $e_0 < e_1 < \dots < e_{n-1}$  in an order preserving way to  $e_0 < e_1 < \dots < e_n$ , with the element  $e_k$  being excluded, and extending affinely. For  $s : S[X]_n$ ,  $n > 0$ , we define  $\partial_k s = s \circ \delta_k : S[X]_{n-1}$ . We get thus maps:

$$S[X]_0 \begin{array}{c} \longleftarrow \\ \longleftarrow \\ \longleftarrow \end{array} S[X]_1 \begin{array}{c} \longleftarrow \\ \longleftarrow \\ \longleftarrow \end{array} S[X]_2 \begin{array}{c} \longleftarrow \\ \longleftarrow \\ \longleftarrow \end{array} S[X]_3 \quad \dots$$

This quickly becomes useful for doing homotopy theory! Using the data provided by the face maps, we can talk about simplices in  $X$  with given boundary. Then, for example, given a map on the boundary of  $\Delta^n$  to  $X$ , we can ask if this map is nullhomotopic, i.e. admits a filler.

The  $\partial_k$  satisfy certain identities. We can think of each  $\partial_k$  as deleting a vertex from a simplex (by passing to the opposite face). Any lower dimensional face is obtained through some sequence of vertex deletions, and we want to assert that the order of these deletions does not matter. For example, suppose that  $k < l$ . Then the codimension 2 face of an  $n$ -simplex that contains neither the  $k$ -th nor the  $l$ -th vertex admits two descriptions: It is the  $k$ -th face of the  $(n-1)$ -simplex that arises as the  $l$ -th face of the original  $n$ -simplex. Or it is the  $(l-1)$ -st face of the  $(n-1)$ -simplex that arises as the  $k$ -th face of the original  $n$ -simplex. Thus:

$$\partial_k \circ \partial_l = \partial_{l-1} \circ \partial_k \quad \text{for } k < l$$

This condition is sufficient to show that there is a unique way of removing any particular set of vertices from a simplex, as the above rule lets us sort repeated application of  $\partial$  to one in which one deletes the vertices in decreasing order.

The collection  $S[X]_\bullet$ , with the information of its face maps, is commonly known as the *singular semi-simplicial set* of  $X$ . This terminology suggests that

each  $S[X]_n$  is usually thought of as a set, as opposed to a space. However, if we think of the collections  $\Delta^n \rightarrow X$  as spaces with the compact-open topology (which will be well behaved since  $\Delta^n$  is an exceptionally nice space), then the face maps  $\partial_k$  are continuous, so this whole story makes sense in the world of topological spaces as well.

Now, instead of talking about points, lines, and triangles in some space  $X$ , we want to move to abstractly talking about some uninterpreted notion of points, lines, and triangles. For the moment, let's work in the world of sets. A **semi-simplicial set** consists of sets  $X_n$  for  $n \geq 0$ , along with face maps  $\partial_k : X_n \rightarrow X_{n-1}$ , for  $k \in \{0, \dots, n\}$ , satisfying the face relations above.

Note that there are lots of semi-simplicial sets that do not arise as the singular semi-simplicial set of some space  $X$ . For example, any singular semi-simplicial set will have a symmetry between lines from  $x$  to  $y$  and lines from  $y$  to  $x$ , whereas this condition is not enforced in the definition of an arbitrary semi-simplicial set. Further, the singular semi-simplicial sets are in fact **simplicial sets** in that they admit a notion of degeneracies that arise from maps  $\Delta^{n+1} \rightarrow \Delta^n$  which collapse two order adjacent vertices. When given abstractly, these degeneracy maps have to satisfy some conditions of their own, as well as conditions giving the boundary of a degenerate simplex. Together, these are called the *simplicial identities*. In general, degeneracies are important because they allow one to talk about boundary preserving homotopies between simplices.


Our goal is to consider semi-simplicial objects in the context of spaces, or, more specifically, in the context of homotopy type theory. The first thing to note is that strict pointwise equality of functions is often the wrong notion to use when talking about spaces; one should instead talk about maps being homotopic. Thus, a semi-simplicial object in the homotopy category of spaces consists of spaces  $X_n$  with (continuous) maps  $\partial_k$  along with homotopies:

$$\alpha_{k,l} : \partial_k \circ \partial_l \simeq \partial_{l-1} \circ \partial_k \quad \text{for } k < l$$


However, this is generally not satisfactory, as for  $k < l < m$ , we can prove that  $\partial_k \circ \partial_l \circ \partial_m \simeq \partial_{m-2} \circ \partial_{l-1} \circ \partial_k$  in two different ways. We require that:

$$\alpha_{k,l} \star \partial_m \cdot \partial_{l-1} \star \alpha_{k,m} \cdot \alpha_{l-1,m-1} \star \partial_k \simeq \partial_k \star \alpha_{l,m} \cdot \alpha_{k,m-1} \star \partial_l \cdot \partial_{m-2} \star \alpha_{k,l}$$

Which we can picture as follows:



However, we're not done yet, because there are new identities for quadruples of deletions. The first identity that comes up is represented by square diagram and says that  $\alpha$  homotopies applied to non-interacting indices commute. The second identity is given one dimension up, and describes a filler for the following figure (called a permutohedron), whose faces are the previously mentioned hexagons and squares. It can be visualised in the following illustration by Tilman Piesk:



Writing down a formula for this is complicated, and things only become worse when you consider sequences of five or more deletions! However, note that, whatever all of these conditions may be, in the case of the singular semi-simplicial object associated to a space  $X$ , all of these coherences can be taken to be identity homotopies, because the results of applying iterated boundaries in different orders are strictly equal. So the most important example is evidently infinitely coherent.

The approach that we have been taking up to this point is known as **fibred**. This is in contrast to working in an **indexed** setting, where we can do away with boundary maps entirely. For example, instead of talking about the collection of lines all together, we talk about separate collections of lines joining specific points  $x$  and  $y$ . Similarly, triangles would be indexed by three points  $x, y, z$  and three lines: from  $x$  to  $y$ , from  $x$  to  $z$ , and from  $y$  to  $z$ .

Now working in the setting of type theory, we define a **semi-simplicial type** to consist of the the following infinite list of data:

```
A : Type
B : A → A → Type
C : {x y z : A} → B x y → B x z → B y z → Type
D : {x y z w : A} {α : B x y} {β : B x z} {γ : B x w} {δ : B y z} {ε : B y w} {ζ : B z w}
  → C α β δ → C α γ ε → C β γ ζ → C δ ε ζ → Type
...
```

Here, we are using Agda curly brace notation to indicate that we are leaving

some of the indices as implicit. We thus write  $C \alpha \beta \in$  instead of  $C x y z \alpha \beta \in$ .

If we had this infinite collection of data, we could sigma-up each of these notions over all of its indices to obtain a total type, as before. We could then define face maps. It is evident that these face maps would be infinitely coherent. Consequently, this makes a very strong case that the definition that we take in the fibred setting should include all possible coherences of any order, and indeed, one can argue that the indexed and fibred definitions are equivalent when all coherences are included in the latter.

For example, given the data  $\partial_0, \partial_1 : X_1 \rightarrow X_0$ , and  $\partial_0, \partial_1, \partial_2 : X_3 \rightarrow X_2$ , we can define:

$$\begin{aligned}\tilde{X}_0 &= X_0 \\ \tilde{X}_1(x, y) &= \sum_{\alpha : X_1} (\partial_1 \alpha \equiv x) \times (\partial_0 \alpha \equiv y)\end{aligned}$$

For the next case, we are defining  $\tilde{X}_2(x, y, z, (\alpha, p_0, q_0), (\beta, p_1, q_1), (\gamma, p_2, q_2))$ .

One may be tempted to say that this consists of  $f : X_2$  with boundary data  $\alpha, \beta, \gamma$ , by asserting, for example, that  $\partial_2 f \equiv \alpha$ . However this equality in  $X_1$  leaves the endpoints free. For example, in the case of the singular semi-simplicial type of a type  $X$ , so long as the lines  $\partial_2 f$  and  $\alpha$  lived in the same connected component of  $X$ , they could be identified by this criterion. We see, then, that this comparison should be performed in the type  $\tilde{X}_1(x, y)$ .

We begin by asking for an element  $f : X_2$  which has the right 0-simplex boundary. This requires the data:  $r_0 : \partial_1 \partial_2 f \equiv x$ ,  $r_1 : \partial_0 \partial_2 f \equiv y$ , and  $r_2 : \partial_0 \partial_1 f \equiv z$ . Now, we can create an element of  $\tilde{X}_1(x, y)$  to compare with  $(\alpha, p_0, q_0)$  via  $(\partial_2 f, r_0, r_1)$ .

In order to create an element of  $\tilde{X}_1(x, z)$ , we want to use  $\partial_1 f$ . We then have that  $r_2 : \partial_0 \partial_1 f \equiv z$ , giving us a proof that the right endpoint is  $z$ . However, for the left endpoint, we only have  $r_0 : \partial_1 \partial_2 f \equiv x$ , and we need to concatenate this on the left with an equality  $\partial_1 \partial_1 f \equiv \partial_1 \partial_2 f$  in order to show that  $\partial_1 \partial_1 f \equiv x$ , as required. A similar analysis applies for  $\tilde{X}_1(y, z)$ . Thus we require the commutation identities  $\alpha_{k,l} : \partial_k \circ \partial_l \equiv \partial_{l-1} \circ \partial_k$  for  $k < l$ .

Provided these identities as part of our starting data, we would then complete the definition as follows:

$$\begin{aligned}\tilde{X}_2(x, y, z, (\alpha, p_0, q_0), (\beta, p_1, q_1), (\gamma, p_2, q_2)) &= \\ \sum_{(f : X_2)} \sum_{(r_0 : \partial_1 \partial_2 f \equiv x)} \sum_{(r_1 : \partial_0 \partial_2 f \equiv y)} \sum_{(r_2 : \partial_0 \partial_1 f \equiv z)} \\ &(\partial_2 f, r_0, r_1) \equiv_{\tilde{X}_1(x, y)} (\alpha, p_0, q_0) \\ &\times (\partial_1 f, \text{apeq } \alpha_{1,2} f \cdot r_0, r_2) \equiv_{\tilde{X}_1(x, z)} (\beta, p_1, q_1) \\ &\times (\partial_0 f, \text{apeq } \alpha_{0,2} f \cdot r_1, \text{apeq } \alpha_{0,1} f \cdot r_2) \equiv_{\tilde{X}_1(y, z)} (\gamma, p_2, q_2)\end{aligned}$$

Using contractible singletons and path algebra, one can show that forming the total spaces of the resulting indexed types leads to types equivalent to  $X_0, X_1, X_2$ . Similarly, starting off with indexed types  $\tilde{X}_0, \tilde{X}_1, \tilde{X}_2$ , forming their total spaces, and then performing the above construction results in equivalent types. This demonstrates an equivalence between the indexed and fibred definitions up to the second stage. One can continue this analysis to the third stage, although writing out the details would be exceptionally painful.

One can ask for a proof of this equivalence in general, which brings us to the punchline: *neither the indexed nor fibred formulations of semi-simplicial types have ever been formally defined inside of type theory!* (As such, the general theorem would involve an equivalence where neither the RHS nor LHS currently have definitions.)

## THE PATTERN

The problem of defining semi-simplicial types (hereafter simply SSTs) in the indexed formulation was posed by Vladimir Voevodsky over ten years ago and has remained one of the most important open problems of homotopy type theory. At first glance, this seems to be an odd state of affairs, because there clearly is a pattern in play. Empirically, you can understand this pattern by writing out the next entry or two; that a line is indexed by its endpoints, and that a triangle is indexed by its bounding lines, clearly generalises.

For this reason, many researchers in homotopy type theory have made their attempts at constructing SSTs, ideally in a proof assistant. As they discover, all direct attacks of the problem are doomed to end in **higher coherence issues**. Despite the relative simplicity of the problem statement in the indexed formulation relative to the fibred formulation, coherences of precisely the same form will eventually come into play.

Of course, just being told that this will be the outcome of your attempt will serve to provide little intuition, and you generally have to go through the process of making your own failed implementation. The reason for this is that the problem appears very tractable at first glance. One thing that may help your intuition in the meanwhile, though, is seeing a precise scientific description, internal to type theory, of the evident pattern.

To see this, we will make all arguments explicit, reorder the terms, and add binary labels:

$A : \text{Type}$

$B : (\underbrace{x : A}_1) (\underbrace{y : A}_{10}) \rightarrow \text{Type}$

$C : (\underbrace{x : A}_1) (\underbrace{y : A}_{10}) (\underbrace{\alpha : B \times x \ y}_{11}) (\underbrace{z : A}_{100}) (\underbrace{\beta : B \times x \ z}_{101}) (\underbrace{\gamma : B \ y \ z}_{110}) \rightarrow \text{Type}$

$$\begin{array}{c}
D : (\underbrace{x : A}_1) (\underbrace{y : A}_{10}) (\underbrace{\alpha : B \times y}_{11}) (\underbrace{z : A}_{100}) (\underbrace{\beta : B \times z}_{101}) (\underbrace{\gamma : B \times y \times z}_{110}) (\underbrace{f_0 : C \times y \times z \times \beta \times \gamma}_{111}) \\
(\underbrace{w : A}_{1000}) (\underbrace{\delta : B \times w}_{1001}) (\underbrace{\epsilon : B \times y \times w}_{1010}) (\underbrace{f_1 : C \times y \times \alpha \times w \times \delta \times \epsilon}_{1011}) (\underbrace{\zeta : B \times z \times w}_{1100}) \\
(\underbrace{f_2 : C \times z \times \beta \times w \times \delta \times \zeta}_{1101}) (\underbrace{f_3 : C \times y \times z \times \gamma \times w \times \epsilon \times \zeta}_{1110}) \rightarrow \text{Type}
\end{array}$$

This ordering of terms is hereby known as **the binary ordering**.

We can think of the canonical  $n$ -simplices as being represented by a string of all 1s. So a point would be 1, a line would be 11, and a triangle would be 111, etc. Then the dependencies of a canonical  $n$ -simplex are given by all non-zero binary arguments strictly less than the indexing number.

We say that the **dimension** of a binary number is one less than the number of 1s that it has. The dimension of a given labelled argument determines whether it indexes a point (dimension 0), or line (dimension 1), or triangle (dimension 2), etc. To understand the ordering that this gives, imagine that you're filling out an infinite dimensional simplex by adding, at each stage, a facet of highest possible dimension with present boundary. The order would go as follows: *point, point, line, point, line, line, triangle, point, line, line, triangle, line, triangle, triangle, tetrahedron, etc.*

Dimensions are not the whole story, though, as, when adding two lines in a row, you have a choice of which line to add first. We say that one binary number is a **subset** of another binary number, if, when aligned to the right, the 1s digits of the first lie in a subset of the position of the 1s digits of the second. The dependencies of a particular argument are precisely those indexed by its non-zero proper subsets, in increasing order. This tells us precisely which facet we're indexing at each stage.


## SIMPLE INVERSE CATEGORIES AND DIAGRAMS

The pattern described above is very satisfying. The question on our plate now is whether or not this observation is sufficient to solve the problem of constructing SSTs. In order to address this, we should scientifically qualify the form of the pattern in play, and then go from there.

The type of each  $A, B, C, \dots$  is given by some telescope mapping to  $\text{Type}$ , and each of the items in the telescope is of the form of one of  $A, B, C, \dots$  with previously occurring variables applied to it. The variables being applied are free of duplicates and are listed in the order in which they occur in the telescope. We also think of each  $A, B, C, \dots$  as having a dimension:  $A$  defines the 0-simplices,  $B$  defines the 1-simplices, etc.


The dependency structure of each telescope can be given by a *dependency list*. A dependency list is either empty, or you can cons on an item to an existing dependency list by specifying on which of the previously occurring items it

depends. For example, the dependency list of a 2-simplex may be pictured as follows:



One will note that a dependency list is precisely equivalent to the data of a *linearised directed acyclic graph*.

Finally, in addition to giving the structure of the dependencies between items in the telescope, we must also label each item with a dimension:



To take an itinerary then, to the type of each of  $A, B, C, \dots$  we associate a particular telescope. (*Or, more precisely, we have a telescope for each dimension.*) The data of each such telescope is precisely specified by a **dimension labelled dependency list**.

In Agda, I have constructed such data corresponding to the pattern seen in the binary ordering of SSTs. Moreover, I have proven the following three properties of this data:

- (a) **Dimension Boundedness:** All items in the labelled dependency list of an  $n$ -simplex have dimension strictly less than  $n$ .
- (b) **Transitivity:** If an item depends on some other item, then it also depends on all of its dependencies.
- (c) **The Shape Property:** If you take an item of dimension  $n$  in some labelled dependency list and *prune* the items preceding it by discarding all items on which it does not depend, then the resulting labelled dependency list is precisely equal to that of the dependencies of an  $n$ -simplex.

Intuitively, for any dimension indexed family of telescopes, you can write a program that will output Agda code that folds up these telescopes (in the form of a postulate context sequentially defining several of the symbols  $A, B, C, \dots$ ). The question is then: *When will the output code be well-typed?*, and the answer is precisely when these three properties hold!

These three properties have important category theoretic significance; they precisely assert that our dimension indexed family of telescopes defines the

structure of a **simple inverse category**<sup>1</sup> (SIC for short). The problem of constructing a corresponding postulate context  $A, B, C, \dots$  is known as that of constructing a family of **diagrams** for this SIC.

We may thus now completely disentangle the problem of constructing SSTs into two steps: First, constructing the corresponding SIC. Second, building the corresponding diagrams.

As alluded to before, the first step has already been completed in Agda. We call the resulting object the *binary SIC*. Everybody's proximal reaction to being introduced to the problem of SSTs is to think that there is an obvious pattern, and the most fitting response to this impression is: *yes, and here is that pattern, formally, in Agda*.

Note that data defining a SIC lives in the world of h-sets (after all, we're working with discrete graphs, with natural number labels, plus some propositions about this data). Diagrams, on the other hand, will involve arbitrary types, hence being the place where the possibility of higher coherence issues creeps in.

Over ten years of failed attempts to construct SSTs suggests to experts in the field that the general problem of building diagrams for all simple inverse categories is impossible in Martin-Löf Type Theory, as with the particular case of the binary SIC.

On the other hand, the problem of constructing diagrams for all simple inverse categories is solvable in MLTT with Axiom K, as truncatedness dispenses of higher coherences. Similarly, without Axiom K, one can construct *semi-simplicial sets* and *semi-simplicial groupoids* (indeed, up to any level of truncatedness, depending on how much rote work one is willing to put into proving coherences).

Additionally, the general problem is also solvable in two level type theory (indeed, this has historically been the main application of the theory). The amount of additional assumptions made in that theory, though, far exceeds the minimal amount of additional proof theoretic strength needed to make the problem tractable, so this state of affairs still leaves the problem of constructing SSTs wide open.

*I claim that a resolution of the problem of constructing SSTs would involve constructing a new type theory, extending MLTT, in which the power added is, in some way, evidently the least amount of power necessary for constructing SSTs. Essentially, what we want to is to come to the crux of the issue, and identify precisely what proof theoretic strength is necessary for having SSTs internal to a type theory. Preliminary work in this direction would then pave a pathway for a later proof of the impossibility of constructing SSTs in MLTT.*

---

<sup>1</sup>We can define a *simple inverse category* to be such a collection of data, except that we also want to quotient out by particular choices of linearisation in the dependency DAG.



## TYPE STRUCTURES

The work that we are about to present emerged from my previous project on formalising the categorical logic of simply typed lambda calculus in Agda. The central definition in that work was that of a *simple contextual category*, which is like a category, except that the notions of *objects* and *morphisms* is derived from more primitive notions of *types* and *terms*.

In simple type theory, there is a notion (i.e. type) of types. We then say that the objects of our theory, *contexts*, are defined to be lists of such types. Upon making these declarations, we have firmly ensconced ourselves in working with a simple, as opposed to dependent, type theory. What then is the structure of types in a dependent type theory?

Egbert Rijke and Paige North have a joint paper on *B-systems*, which are an adequate model of dependent type theories internal to MLTT. In this paper, the structure of types in a dependent type theory is specified by giving the collections of length  $n$  contexts, along with fundamental context projections:

$$\{\star\} = E_0 \xleftarrow{\partial} E_1 \xleftarrow{\partial} E_2 \xleftarrow{\partial} E_3 \xleftarrow{\partial} \dots$$

Egbert and I were independently working on modelling dependent type theory in Agda, with my eventual goal being optimising for ergonomics in formalising metatheory. The formulation above is fibred, as opposed to indexed, and is thus essentially unsuitable for formalisation (for example, saying that you have a type in a context involves postulating and transporting around equalities). Egbert and I independently arrived at using an equivalent indexed formulation in our code, using inductive types, which we shall call a **type structure**, or **TyStr**, for short:

```
codata TyStr where
  ty  : TyStr → Type
  ex  : (T : TyStr) → ty T → TyStr
```

In words, a TyStr consists of a notion of types, and for every such type, another TyStr representing the types depending on it. From a TyStr, we are able to define, using induction-recursion, the types of length  $n$  contexts:

```
data Ctx (T : TyStr) (n : ℕ) where
  ∅  : Ctx zero
  _+_ : {n : ℕ}
    (Γ : Ctx T n) → ty (exs T Γ) → Ctx T (suc n)

exs : (T : TyStr) {n : ℕ} → Ctx T n → TyStr
exs T ∅ = T
exs T (Γ + A) = ex (exs T Γ) A
```

The above definition say that, to extend a context  $\Gamma$ , you add a type depending on  $\Gamma$ . In order to make sense of types depending on a context, this has to be defined mutually with an operation of extending a type structure by a context.

There is a neat way of thinking about TyStrs: We define the *linear SIC* to consist of the  $\mathbb{N}$  indexed family of labelled dependency lists where the  $n$ -th list consists on  $n$ -items of increasing dimensions, and such that each item depends on all of the previous items. If we built a postulate context out of the linear SIC, it would look like:

```
A : Type
B : (x : A) → Type
C : (x : A) (y : B x) → Type
D : (x : A) (y : B x) (z : C x y) → Type
...
```

Then the type of data of defining each element of the infinite list  $A, B, C, \dots$  is exactly given by the type of TyStrs. Thus, in a slightly different sense than how we were describing the problem before, the type TyStr is a construction of the type of diagrams for the linear SIC.

In the case of TyStrs, the coinductive formulation is really expressing something profound about the type of linear diagrams that we do not directly see in the fibred construction described above. We can express this as a mapping-in universal property:

```
UP : (A : Type) (Z : A → Type)
    (S : (x : A) → Z x → A) → (A → TyStr)
ty  (UP A Z S x) = Z x
ex  (UP A Z S x) x' = UP A Z S (S x x')
```

Thus, to construct a function  $A \rightarrow \text{TyStr}$ , you need to give a *ZS structure* on the type  $A$ . This consists of  $Z$ , a type dependent on  $A$ , and  $S$ , a family of functions  $Z\ x \rightarrow A$ . Note that the types of  $Z$  and  $S$  are read directly off of the fields of TyStr. With this data, every element of  $A$  defines a TyStr.

We have shown that the problem of constructing the type of diagrams of a SIC can be solved in certain cases. In particular, if the type of diagrams happens to have a universal property, then the question about whether or not we can construct such a type comes down to being able to construct anything satisfying that universal property (because universal properties characterise things uniquely). We can prove that the fibred formulation above satisfies the UP of linear diagrams, so it's possible to perform this construction without coinduction. However, the universal property in question is manifestly coinductive so having coinduction is imperative for having a canonical type of linear diagrams.

*The plan, therefore, is exhibit a universal property associated to SSTs, and to construct a type theory capable of expressing that universal property.*

## SSTs ARE A DEPENDENT TYPE THEORY

The problem of constructing SSTs has traditionally been posed as that of constructing a function  $SST : \mathbb{N} \rightarrow \text{Type}$  such that  $SST\ n$  is the type of  $n$ -truncated SSTs (this is usually thought of as a sigma type specifying the first  $n + 1$  of  $A, B, C, \dots$ ). A solution to this problem would lead to a definition of infinite SSTs, albeit in a messy way.

A refinement on the problem statement is to instead ask for  $SST : \text{TyStr}$ . In other words,  $SST$  naturally has the structure of a  $\text{TyStr}$ . We could manually define the first few stages of this  $\text{TyStr}$  by copattern matching:

```
SST : TyStr
ty SST = Type
ty (ex SST A) = A → A → Type
ty (ex (ex SST A) B) =
  (x y z : A) → B x y → B x z → B y z → Type
ex (ex (ex SST A) B) C = ?
```

If one had such a  $\text{TyStr}$ , then  $n$ -truncated SSTs would be given by the type  $Ctx\ SST\ n$ , and infinite SSTs would analogously be infinite contexts – here  $Ctx^\infty$  can be defined coinductively in a straightforward way. Moreover, the general problem of constructing diagrams is best phrased as that of turning a SIC into a  $\text{TyStr}$ .

Surprisingly, this formulation of the problem is **not** the one that we work towards in the remainder of this note. The reason for this is that the type of diagrams  $Ctx^\infty\ SST$  does not have a universal property on its own, and we must instead work in a more general setting.

The plan is that we will describe  $SST : \text{TyStr}$  such that  $ty\ SST$  is the type of infinite SSTs. Intuitively, then, we think of  $A : ty\ SST$  as consisting of the following infinite collection of data:

```
A0 : Type
A1 : A0 → A0 → Type
A2 : {x y z : A0} → A1 x y → A1 x z → A1 y z → Type
...
```

If we accept this definition, for the time being, then we must next ask for the interpretation of  $B : ty\ (ex\ SST\ A)$ . This will consist of the data:

```
B0 : A0 → Type
B1 : {x y : A0} → A1 x y → B0 x → B0 y → Type
B2 : {x y z : A0} {α : A1 x y} {β : A1 x z} {γ : A1 y z} {x' : B0 x} {y' : B0 y} {z' : B0 z}
  → A2 α β γ → B1 α x' y' → B1 β x' z' → B1 γ y' z' → Type
...
```

B is known as a dependent semi-simplicial type over A.  $A_0$  was just the type of points of A, whereas  $B_0 x$  is the type of points of B living over some point  $x : A_0$ . Similarly, the type  $B_1 \alpha x' y'$  describes lines in B living over  $\alpha : A_1 x y$ , starting from  $x' : B_0 x$ , and ending at  $y' : B_0 y$ . In general, the type of dependent n-simplices in B will be indexed by a filled-in n-simplex in A, as well as a lift of its boundary to B.

Similarly, we identify  $C : ty (ex (ex SST A) B)$  with the data:

$$\begin{aligned} C_0 &: \{x : A_0\} \rightarrow B_0 x \rightarrow \text{Type} \\ C_1 &: \{x y : A_0\} \{\alpha : A_1 x y\} \{x' : B_0 x\} \{y' : B_0 y\} \\ &\rightarrow B_1 \alpha x' y' \rightarrow C_0 x' \rightarrow C_0 y' \rightarrow \text{Type} \\ &\dots \end{aligned}$$

The pattern is much the same: a doubly dependent n-simplex in C is indexed by a filled in dependent n-simplex in B, as well as a lift of its boundary to C.

We have defined  $SST : TyStr$  to capture SSTs of all levels of dependency. The remarkable observation is that SST defines the collection of types in a model of dependent type theory, in the sense of B-systems. To show this, we have to construct weakening, elements, substitution, and variables, and show that these constructs satisfy a list of properties.

The most interesting construct is that of elements. Intuitively, we can consider a notion of morphisms of SSTs, and elements of  $A : ty SST$  should be equivalent to morphisms from the terminal SST to A. The terminal SST has exactly one filler for any boundary data, and so a morphism would send  $\star_0 : \top_0$ , to  $t_0 : A_0$ . It would then have to send the only 1-simplex,  $\star_1 : \top_1 \star_0 \star_0$  to some  $t_1 : A_1 x_0 x_0$ . Similarly, it would have to send the only 2-simplex,  $\star_2 : \top_2 \star_1 \star_1 \star_1$ , to some  $t_2 : A_2 x_1 x_1 x_1$ . Thus, an element of  $A : ty SST$  consists of the infinite collection of data:

$$\begin{aligned} t_0 &: A_0 \\ t_1 &: A_1 t_0 t_0 \\ t_2 &: A_2 t_1 t_1 t_1 \\ &\dots \end{aligned}$$

An element of  $B : ty (ex SST A)$  should correspond to a dependent morphism of SSTs from A to B (in other words, a section). Thus for every  $x : A_0$ , we get some  $s_0 x : B_0 x_0$ , and for every  $\alpha : A_0 x y$ , we would get some  $s_1 \alpha : B_1 \alpha (s_0 x) (s_0 y)$ . Thus, an element of  $B : ty (ex SST A)$  consists of the infinite collection of data:

$$\begin{aligned} s_0 &: (x : A_0) \rightarrow B_0 x \\ s_1 &: \{x y : A_0\} (\alpha : A_1 x y) \rightarrow B_1 \alpha (s_0 x) (s_0 y) \\ s_2 &: \{x y z : A_0\} \{\alpha : A_1 x y\} \{\beta : A_1 x z\} \{\gamma : A_1 y z\} (f : A_2 \alpha \beta \gamma) \\ &\rightarrow B_2 f (s_1 \alpha) (s_1 \beta) (s_1 \gamma) \\ &\dots \end{aligned}$$

The most important part of the B-system structure on SST, for our purposes, is the weakening structure. Formally, we define:

```
codata TyMor ( $\mathcal{T} \ \mathcal{S} : \text{TyStr}$ ) where
   $fun : \text{TyMor } \mathcal{T} \ \mathcal{S} \rightarrow ty \ \mathcal{T} \rightarrow ty \ \mathcal{S}$ 
   $up : (\mathfrak{f} : \text{TyMor } \mathcal{T} \ \mathcal{S}) \ (A : ty \ \mathcal{T})$ 
     $\rightarrow \text{TyMor } (ex \ \mathcal{T} \ A) \ (ex \ \mathcal{S} \ (fun \ \mathfrak{f} \ A))$ 

codata WkStr ( $\mathcal{T} : \text{TyStr}$ ) where
   $wk : \text{WkStr } \mathcal{T} \rightarrow (A : ty \ \mathcal{T}) \rightarrow \text{TyMor } \mathcal{T} \ (ex \ \mathcal{T} \ A)$ 
   $up : \text{WkStr } \mathcal{T} \rightarrow (A : ty \ \mathcal{T}) \rightarrow \text{WkStr } (ex \ \mathcal{T} \ A)$ 
```

In the fibred formulation of the type of linear diagrams, a TyMor would be equivalent to a natural transformation of such diagrams. The key point is that TyMors define not only how to operate on top-level types, but also on arbitrary contexts. A WkStr on  $\mathcal{T}$  then is a way to shift contexts in  $\mathcal{T}$ , or any of its iterated extensions, over by a type in what we conceptually view as adding a trivial dependency. Weakening on SST is intuitively constructed in this way – we weaken by adding trivial dependencies.

## GENERALISED (CO)INDUCTION

Let’s consider first the case of type streams. We define:

```
codata TyStream where
   $head : \text{TyStream} \rightarrow \text{Type}$ 
   $tail : \text{TyStream} \rightarrow \text{TyStream}$ 
```

Thus a TyStream is just an infinite list (aka stream) of types. A morphism of TyStreams is defined in the evident way as “*rungs on a ladder*”. Note that if we had streams of natural numbers, there would be no similar suitable notion of morphisms, so TyStreams are a special kind of stream.

Now, what would it mean to *inductively define a TyStream*? In other words, we want to make sense of definitions of the form:

```
data Foo : TyStream where
   $Z : head \ Foo$ 
   $S : Foo \rightarrow tail \ Foo$ 
```

This is written in the internal logic of the category of TyStreams, so, for example, S is a morphism of TyStreams.

Well, in this example, we ask ourselves whether this definition meaningfully defines  $Foo : \text{TyStream}$ ? Well, we have an element  $Z : head \ Foo$ , and a way to promote elements down the stream, so this should meaningfully define something equivalent to a stream of units.

When the possibility of such generalised TyStream-valued induction was proposed, Mike and I immediately wrote down a preliminary definition of SSTs:

```
codata SST : TyStr where
  Z : SST → Type
  S : (X : SST) → Z X → ex SST X
```

This is TyStr-valued coinduction, and is written in the internal language of TyStrs. *There is a lot to unpack here*, although the main idea is that the *type* of infinite SSTs does not on its own have a universal property, instead, you have to generalise to all notions of dependency, and only then does the TyStr of SSTs acquire a universal property.

We then spent the next seven months thinking about this fictional code fragment and working on a type theory in which it would type-check. In general, the original definition above is almost right, but what one realises after working on this for several weeks is that the setting has to be expanded to TyStrs with weakening in order to recover simplex types. Conjecturally, an analogous B-system-valued coinduction will correctly recover the B-system of SSTs described above, with extraction algorithms for both simplex types and element components.

## UNPACKING THINGS

We begin with the projection  $Z : \text{SST} \rightarrow \text{Type}$ . This should be a TyMor, so Type should be a TyStr. We can define the universal TyStr as follows:

```
u' : Type → TyStr
ty (u' A) = A → Type
ex (u' A) B = u' (Σ A B)

u : TyStr
ty u = Type
ex u A = u' A
```

An alternative definition is given as follows:

```
u' : Type → TyStr
u' A = UP Type (λ A → A → Type) Σ

u : TyStr
u = u' ⊤
```

In other words,  $A : \text{ty } u$  is a type,  $B : \text{ty } (ex\ u\ A)$  is a type depending on  $A$ , and  $C : \text{ty } (ex\ (ex\ u\ A)\ B)$  is a type depending on  $\Sigma A\ B$ , etc. Thus, morally,  $u$  is the TyStr of types, and a TyMor  $S : \mathcal{T} \rightarrow u$  should represent a dependent type structure over  $\mathcal{T}$ .

It is often easier to work with dependent type structures directly. We define:


**codata**  $\text{TyStr}^d (\mathcal{T} : \text{TyStr})$  **where**  
 $ty^d : \text{TyStr}^d \mathcal{T} \rightarrow ty \mathcal{T} \rightarrow \text{Type}$   
 $ex^d : (\mathcal{S} : \text{TyStr}^d \mathcal{T}) (A : ty \mathcal{T}) \rightarrow ty^d \mathcal{S} A$   
 $\rightarrow \text{TyStr}^d (ex \mathcal{T} A)$

One also obtains a notion  $Ctx^d \mathcal{S} \Gamma$  of dependent contexts in  $\mathcal{S}$  living over some  $\Gamma : Ctx \mathcal{T} n$ . In the same way that the fibred notion of linear diagrams is characterised by giving all of the length  $n$  contexts, along with their projections, the fibred notion of dependent linear diagrams is given by all of the  $\Gamma$ -indexed contexts, and their projections.

We thus think of  $Z$  as a  $\text{TyStr}^d$  over  $\text{SST}$ . The projection  $S : (X : \text{SST}) \rightarrow ZX \rightarrow ex \text{SST } X$  is then to be thought of as a morphism between two dependent type structures over  $\text{SST}$ . We thus need to define, for every type structure  $\mathcal{T}$ , a dependent type structure  $EX \mathcal{T} : \text{TyStr}^d \mathcal{T}$ .

We will now introduce a notation for contexts. If  $\mathcal{T}$  is a  $\text{TyStr}$ , we will write  $\Gamma = A \rightarrow B \rightarrow C \rightarrow D : Ctx \mathcal{T} 4$  to mean that  $A : ty \mathcal{T}$ , and that  $B$  depends on  $A$ , etc. Further, if  $\mathcal{S} : \text{TyStr}^d \mathcal{T}$ , we will write  $\tilde{\Gamma} = A' \Rightarrow B' \Rightarrow C' \Rightarrow D' : Ctx^d \mathcal{S} \Gamma$  to indicate that  $A' : ty^d \mathcal{S} A$ ,  $B' : ty^d (ex^d \mathcal{S} A A')B$ , etc. Thus, in order to define a dependent type structure over  $\mathcal{T}$ , we must give meaning to dependent contexts of the form  $\tilde{\Gamma} = A' \Rightarrow B' \Rightarrow C' \Rightarrow D' : Ctx^d \mathcal{S} \Gamma$ .

Now, it turns out that in order to define  $EX \mathcal{T}$ , we need to assume that  $\mathcal{T}$  is equipped with a weakening structure. Under this assumption, the dependent contexts  $\tilde{\Gamma} = A' \Rightarrow B' \Rightarrow C' \Rightarrow D' : Ctx^d \mathcal{S} \Gamma$  living over  $\Gamma = A \rightarrow B \rightarrow C \rightarrow D : Ctx \mathcal{T} 4$  are described as consisting of the data of  $A', B', C'$ , and  $D'$  fitting into the following diagram (called the *EX* diagram):



Given a context  $\Gamma : Ctx \text{SST } n$ , we refer to a dependent context  $\tilde{\Gamma} : Ctx^d Z \Gamma$  as a  $Z$ -section of  $\Gamma$ .  $S$  is therefore a gadget that, given any context  $\Gamma : Ctx \text{SST } n$ , transforms  $Z$ -sections of  $\Gamma$  into sections of  $EX \text{SST}$  over  $\Gamma$ .

## SIMPLEX EXTRACTION

Suppose that we have  $\mathcal{T} : \text{TyStr}$ , equipped with a  $\mathcal{W} : \text{WkStr } \mathcal{T}$ . Then a ZS-structure on  $(\mathcal{T}, \mathcal{W})$  consists of  $Z : \text{TyStr}^d \mathcal{T}$ , and  $S$ , a morphism of  $\text{TyStr}^d$ s over  $\mathcal{T}$  from  $Z$  to  $EX \mathcal{T}$ .

From the previous discussion, this data should be sufficient to define a  $\text{TyMor } \mathcal{T} \rightarrow \text{SST}$ . So every  $A : ty \mathcal{T}$  should give rise to an infinite semi-simplicial type, and every  $B : ty (ex \mathcal{T} A)$  should give rise to an infinite semi-simplicial type depending on the previous one, etc. In order to see that this is the case, we present an algorithm for extracting simplex types  $A_0, A_1, A_2, \dots$  from  $A$ .

Given  $A : ty \mathcal{T}$ , this can be viewed as the length one context

$$\Gamma_0 = \emptyset + A : Ctx \mathcal{T} 1.$$

The type  $A_0$  of 0-simplices is given as  $ty^d Z A : \text{Type}$ . If we now choose a 0-simplex  $x : ty^d Z A$ , then we obtain a Z-section  $\emptyset + x : Ctx^d Z \Gamma_0$ . Applying  $S$  to this Z-section turns it into  $\emptyset + S x : Ctx^d (EX \mathcal{T}) \Gamma_0$ .

Now, the key thing to note is that length  $n$  contexts in  $EX \mathcal{T}$  translate into length  $2n$  contexts in  $\mathcal{T}$ , given by the outer stair shaped sequence of dependencies in the  $EX$  diagram. Thus, what we really have now is a context:

$$\Gamma_1 = \emptyset + A + S x : Ctx \mathcal{T} 2.$$

We can now form a Z-section of this context:  $\emptyset + y + \alpha : Ctx^d Z \Gamma_1$ . Note that  $y : ty^d Z A$ , so it's a 0-simplex as well, and then  $\alpha : ty^d (ex^d Z A y) (S x)$  is the type of 1-simplices whose boundary consists of the points  $x$  and  $y$ . We then apply  $S$  to this to obtain the length 4 context:

$$\Gamma_2 = \emptyset + A + S y + W_{S y}(S x) + S \alpha : Ctx \mathcal{T} 4.$$

We then take another Z-section, this time of  $\Gamma_2$ :  $\emptyset + z + \beta + \gamma + f_0 : Ctx^d Z \Gamma_2$ . We have that  $z : ty^d Z A$  and  $\beta : ty^d (ex^d Z A z) (S y)$ , so  $z$  is a 0-simplex and  $\beta$  is a 1-simplex from  $y$  to  $z$ . The difficulty, however, is that

$$\gamma : ty^d (ex^d (ex^d Z A z) (S y) \beta) (W_{S y}(S x)).$$

What we want is for this type to be  $ty^d (ex^d Z A z) (S x)$ , in other words, that of a 1-simplex from  $x$  to  $z$ . The appearance of  $\beta$  in the above expression suggests that the type of  $\gamma$  depends on  $\beta$ , which is incorrect, however, there is also a weakening, and the idea is that this weakening should cancel the extension, giving us the desired reduction.

Ignoring this problem for the moment, we obtain that

$$f_0 : ty^d (ex^d (ex^d (ex^d Z A z) (S y) \beta) (W_{S y}(S x)) \gamma) (S \alpha),$$



and this should be the type of 2-simplices with boundary  $x, y, z, \alpha, \beta, \gamma$ . In the next iteration, we would apply  $S$  to this context to obtain a length 8 context.  $Z$  sectioning this context would pass from the 2-simplex that we previously specified to a 3-simplex. The type of the last element of that  $Z$ -section would be the type of 3-simplices. We can repeat this process to extract all of the simplex types.

Fascinatingly, the above algorithm actually describes how to extract simplex types from types at all levels of dependency in  $\mathcal{T}$ . Instead of starting with the context  $\emptyset + A$ , we could have started with the context  $\emptyset + A + B$ . A  $Z$ -section of this context would be  $\emptyset + x + x'$ , where  $x : ty^d Z A$  is a 0-simplex in  $A$ , and  $ty^d (ex Z A x) B$  is the type of dependent 0-simplices in  $B$  living over  $x$ . Repeatedly applying  $S$  and sectioning the new context would yield, as the type of the last element, all of the dependent simplex types in  $B$ .

In Agda, I have formalised the notion of a  $ZS$ -structure on a  $TyStr$  with weakening. From any such data, given  $A : ty \mathcal{T}$ , I construct a derived  $TyStr$  such that its contexts look like  $\emptyset + x + y + \alpha + z + \beta + \gamma + f_0$  as above. In other words, taking contexts in this  $TyStr$  achieves the process of filling out successive facets of an infinite-dimensional simplex. (Producing such a gadget is my preferred way of solving the simplex extraction problem.)

This Agda file solves the problem of producing *unreduced simplex types*, in the sense of the difficulty that we alluded to above. Since we are working in the theory of type structures with weakening, the correct notion of morphism involves a weakening preservation condition. The universe  $\mathcal{U}$  has a weakening structure that just weakens a type by adding a trivial dependency, so  $Z$  preserving weakening would be a proposition that would allow us to cancel the  $ex$  with the weakening in the type of  $\gamma$ . In the type of a 3-simplex, as well, we would see weakening under  $S$ , and, in order to get reduced types, we also have to postulate that  $S$  preserves weakening – in effect, one weakening inside of  $S$  moves to two weakenings outside of  $S$ . (Also we will need weakening to preserve weakening in order to be able to state  $S$  preservation.) In experiments with rewriting, we observe that if all of these properties were definitional, then we would have reduced simplex types on the nose. Thus the entire game of the current research is making sense of this.

Finally, note that the order in which we extracted the simplices does not match the binary ordering since, for example, the line from  $y$  to  $z$  came before the line from  $x$  to  $z$  (the ordering of the dimensions, however, is a match). In the *EX* diagram, we chose to have the squiggly arrows indicating weakening be horizontal, which resulted in the length  $2n$  context having a staircase shape. We could have instead drawn the diagram with vertical squiggly arrows, which would result in the length  $2n$  context having an L-shape. The first version has an advantage in that adding in that insertions to the stair shaped context occur at the end, unlike the L-shaped context, in which one of the insertions occurs at the middle. The remarkable observation about the second version,

though, is that the same simplex extraction process as described above would yield simplices ordered *exactly in the binary ordering!*

## THE SINGULAR SEMI-SIMPLICIAL TYPES

In certain cases, we will be able to construct ZS-structures in which the weakening preservation laws hold definitionally. When this happens, the simplex extraction algorithm above will yield “*reduced*” expressions, and the problem of constructing that family of SSTs will have a complete resolution in present-day Agda. Fortunately, in the case of the most important example, the singular semi-simplicial types, we are able to produce such a structure!

Suppose that we have a type  $X$ , and we want to produce the SST  $\text{Sing } X$ . We have a mapping-in universal property that lets us construct SSTs out of types in some type structure  $\mathcal{T}$ . What should this  $\mathcal{T}$  be?

One observation to make is that  $\mathcal{T}$  cannot be trivial in several senses. First,  $\mathcal{T}$  cannot be the type structure with only a single closed type and no dependent types living over it. This is because  $\mathcal{T}$  has to admit a weakening structure, so we must have, in the least, all possible weakenings of any closed type. We might then be tempted to take  $\mathcal{T}$  to be the next simplest case – one in which there is exactly one type at each level of dependency (so units all the way up). Unfortunately, in that case, every dependent type will be a weakening of the unique closed type, and weakening preservation on  $Z$  would imply that  $Z$  is essentially constant. For example, the type of points would equal the type of lines between any  $x$  and  $y$ , as well as the type of triangle with any given boundary. This does not generally hold in the case of singular SSTs, or indeed, in most examples of SSTs, we thus see that the choice of  $\mathcal{T}$  has to be uniquely tailored to the situation.

The key is to generalise the problem of constructing  $\text{Sing } X$  for a particular type  $X$  to that of constructing  $\text{Sing}$  to any type, and indeed, to any dependent type at all levels of dependency. In other words, we will take  $\mathcal{T}$  to be  $\mathcal{U}$ , the universal type structure constructed above.

We construct the rest of the structure as follows:

- Weakening of types behaves as one would expect.
- We define  $Z$  to be the *tautological dependent type structure on  $\mathcal{U}$* . So, for example, suppose that  $\Gamma = \emptyset + A + B + C : \text{Ctx } \mathcal{U} \ 3$ . We then have  $A : \text{Type}$ ,  $B : A \rightarrow \text{Type}$ , and  $C : \Sigma A \ B \rightarrow \text{Type}$ , and a  $Z$ -section  $\emptyset + x + y + z : \text{Ctx}^d \ Z \ \Gamma$  consists of  $x : A$ ,  $y : B \ x$ , and  $z : C \ (x, y)$ .
- In defining  $S$ , if we have  $A : \text{Type}$  and  $x : A$ , then  $S \ x$  has type  $A \rightarrow \text{Type}$ . We define this to be  $\lambda y \rightarrow x \equiv y$ . For the higher cases, we use some cleverness involving cubical dependent path types.

With this data, every type  $X : \text{Type}$  should define a SST  $\text{Sing } X$ . Applying the extraction algorithm yields:

$x : X$	$w : X$
$y : X$	$\delta : z \equiv w$
$\alpha : x \equiv y$	$\epsilon : y \equiv w$
$z : X$	$f_1 : \text{PathP } (\lambda i \rightarrow y \equiv \delta i) \beta \epsilon$
$\beta : y \equiv z$	$\zeta : x \equiv w$
$\gamma : x \equiv z$	$f_2 : \text{PathP } (\lambda i \rightarrow y \equiv \delta i) \gamma \zeta$
$f_0 : \text{PathP } (\lambda i \rightarrow x \equiv \beta i) \alpha \gamma$	$f_3 : \text{PathP } (\lambda i \rightarrow y \equiv \epsilon i) \alpha \zeta$
$\Delta_0 : \text{PathP } (\lambda i \rightarrow \text{PathP } (\lambda j \rightarrow x \equiv f_1 i j) \alpha (f_2 i)) f_0 f_3$	

These types were generated by Agda, thus demonstrating that our universal property can handle the most important example. To my knowledge, this is the first ever construction of singular SSTs within type theory.